/*
 * Licensed under the MIT License
 * 
 * Copyright (c) 2011 Kay Kreuning
 * 
 * Permission is hereby granted, free of charge, to any person obtaining a copy
 * of this software and associated documentation files (the "Software"), to deal
 * in the Software without restriction, including without limitation the rights
 * to use, copy, modify, merge, publish, distribute, sublicense, and/or sell
 * copies of the Software, and to permit persons to whom the Software is
 * furnished to do so, subject to the following conditions:
 * 
 * The above copyright notice and this permission notice shall be included in
 * all copies or substantial portions of the Software.
 * 
 * THE SOFTWARE IS PROVIDED "AS IS", WITHOUT WARRANTY OF ANY KIND, EXPRESS OR
 * IMPLIED, INCLUDING BUT NOT LIMITED TO THE WARRANTIES OF MERCHANTABILITY,
 * FITNESS FOR A PARTICULAR PURPOSE AND NONINFRINGEMENT. IN NO EVENT SHALL THE
 * AUTHORS OR COPYRIGHT HOLDERS BE LIABLE FOR ANY CLAIM, DAMAGES OR OTHER
 * LIABILITY, WHETHER IN AN ACTION OF CONTRACT, TORT OR OTHERWISE, ARISING FROM,
 * OUT OF OR IN CONNECTION WITH THE SOFTWARE OR THE USE OR OTHER DEALINGS IN
 * THE SOFTWARE.
 * 
 * http://code.google.com/p/hyves-as3-sdk/
 */
package nl.hyves.api.net
{
	import com.adobe.cairngorm.contract.Contract;
	import nl.hyves.api.core.IHyvesAPI;
	import nl.hyves.api.utils.AuthUtil;

	import flash.events.Event;
	import flash.events.IOErrorEvent;
	import flash.net.URLLoader;

	public class HyvesRequest implements IHyvesRequest
	{
		protected var service:IHyvesAPI;
		protected var _parameters:Object;
		protected var callback:Function;
		protected var loader:URLLoader;

		public function get parameters():Object
		{
			return _parameters;
		}

		public function HyvesRequest(service:IHyvesAPI, parameters:Object, callback:Function)
		{
			Contract.precondition(service != null);
			Contract.precondition(parameters != null);
			Contract.precondition(callback != null);
			
			this.service = service;
			_parameters = parameters;
			this.callback = callback;
			
			Contract.postcondition(this.service != null);
			Contract.postcondition(this.parameters != null);
			Contract.postcondition(this.callback != null);
		}

		public function load():void
		{
			loader = new URLLoader();
			loader.addEventListener(Event.COMPLETE, onComplete);
			loader.addEventListener(IOErrorEvent.IO_ERROR, onIOError);
			loader.load(AuthUtil.buildRequest(parameters, service.endpoint, service.consumer, service.accesstoken));
		}

		protected function onComplete(event:Event):void
		{
			removeListeners();

			callback(new HyvesResponse(event.target.data));
		}

		protected function onIOError(event:IOErrorEvent):void
		{
			removeListeners();
		}

		protected function removeListeners():void
		{
			if (loader.hasEventListener(Event.COMPLETE))
			{
				loader.removeEventListener(Event.COMPLETE, onComplete);
			}

			if (loader.hasEventListener(IOErrorEvent.IO_ERROR))
			{
				loader.removeEventListener(IOErrorEvent.IO_ERROR, onIOError);
			}
		}
	}
}
